
IchigoJam初級ゲームプログラミング  ～サウンド編～

【講師】ロボット工房 RoBoCoBo　中野 司・佐藤 誉夫 

【会場】北九州パレス(勤労青少年文化センター) 2F 

【日時】12/24(月・祝) 10:00～15:00 (昼休み12:00-13:00)

０．はじめに 

 ◆コース概要

　- ねらい

　　・プログラムを「書く」

　　・コンピュータ、わかる

　　・頭、よくなる

　　・英語になれる 

　　・国産イチゴを守る 

　- 本日のスケジュール

　　　0. はじめに

　　　1. 序章 - 音を鳴らそう

　　　2. ［アプリNo.110］ジューク・ボックス

　　　3. ［アプリNo.111］電子ピアノ

　　　4. ［アプリNo.112］リプレイ（順番当てゲーム）

　　　5. ［アプリNo.113］ピアノの達人（キー押しゲーム）

　　　6. ［アプリNo.110］アプリ・セレクター

　　　7. おわりに



 

  /  3 40

◆スタートアップおさらい 

　- コマンドとプログラム 

　- プログラム編集と入力テクニック

　　・スクロールアウトしたら、さようなら…またLISTしてね

　　・プログラム行のコピーと削除

　　・もう忘れて… NEW 

　- LISTとスクロール停止

　　・F4キー　→すぐに→　ESCキー

　　・LIST-100 … ～100行目まで

　　・LIST100,0 … 100行目～

　- ファイル表示（FILES）と保存（SAVE 0～3）

　　・プログラム１行目はコメントにしておく … 10 'ﾌﾟﾛｸﾞﾗﾑ ﾒｲ 

◆メモリ容量 (1kB ×４バンク)

　- スペースは省略可

　- 空きメモリは ? FREE() … F6キー

　- LOAD と LRUN（他バンクを LOAD して RUN）

 ◆EEPROM (64kB = 1kB ×64バンク)

　- EEPROMの差し込む向きに注意！ 

　 

　- メモリバンク 0～3 を研究用に空けておこう

　- EEPROMバンク 100～109 も君たちが使おう

　- 全ファイル表示（FILES 0）と保存（SAVE 100～163 または 227）

　- 今回のファイル表示コマンド FILES113（FILES 0,113 に同じ）

 ◆キーボード操作

　- 小文字

　- カナ・特殊文字

１．序章 - 音を鳴らそう 

 ◆圧電サウンダー接続方法　

　- 圧電サウンダーの両足をそれぞれ次の穴に差し込んでください。

　　右列、上から４番目のSND ピン　

　　右列、上から７番目のGNDピン
　　＊極性はありません。

◆BEEP命令 

◆PLAY命令とMML

　- なんか音痴…

２．ジューク・ボックス 

◆文字コード／キーコードを表示させよう
―――――――――――――――――――― 

1 'JukeBox 0 

200 K=INKEY() 

210 ?"KEY CODE = ";K 

300 WAIT 5: GOTO 200

――――――――――――――――――――

プログラム名

押されたキーを読み取り、Kに代⼊

Kに⼊っている数字を表⽰

5/60秒待つ  ２００⾏に⾏く

　- NEW してプログラム入力 

　- キー入力命令 INKEY()

　- GOTO 行番号

　- SAVE 110 で保存	

◆曲を選んで演奏させよう 

―――――――――――――――――――― 
1 'JukeBox 1 

10 A="T60 <BGAD2.R DABG2.R" 

20 B=“T180<G.G8A2G2<C2>B2.R 

G.G8A2G2<D2C2.>RG.G8<G2E2C. 

C8>B2A2.R <F.F8E2C2D2C2.R" 

30C=“T200<D#8C#8>F#8R8<F#F# 

D#8C#8>F#8R8<F#F#D#8C#8>F#8 

R8<F#>D#<F#>C#<FFD#8C#8>C#8 

R8<FFD#8C#8>C#8R8<FFD#8C#8 

>C#<F>D#<F>F#<F#F#" 

40　D=“T180<EEE2EEE2EGC.D8E2 
R2FFF.F8FEE.E8EDDCD2GR" 

50 E=“<G<CC8D8C8>B8AAA<DD8 

E8D8C8>BGG<EE8F8E8D8C>AG8 

G8A<D>B<C" 

60 F=“<<C>B8.A16G.F8EDC.G8 

A.A8B.B8<C.” 

100 CLS 

110 ?"Ǯ1 ﾁｬｲﾑ” 

120 ?"Ǯ2 ﾊｯﾋﾟｰﾊﾞｰｽﾃﾞｰ" 

130 ?"Ǯ3 ﾈｺ ﾌﾝｼﾞｬｯﾀ....” 

140 ?"Ǯ4 ｼﾞﾝｸﾞﾙ ﾍﾞﾙ" 

150 ?"Ǯ5 ﾒﾘｸﾘ ｵﾒﾃﾞﾄ" 

160 ?"Ǯ6 ﾓﾛﾋﾞﾄ ｺｿﾞﾘﾃ"  

200 @SONG: K=INKEY() 

210 IF K=ASC("1") PLAY A 

220 IF K=ASC("2") PLAY B 

230 IF K=ASC("3") PLAY C 

240 IF K=ASC("4") PLAY D 

250 IF K=ASC("5") PLAY E 

260 IF K=ASC("6") PLAY F 

300 WAIT 5: GOTO @SONG 

――――――――――――――――――――

曲データ

画⾯クリアコマンド

曲⽬表⽰

演奏コマンド

@SONGへ⾏く

　- 文字コード取得命令 ASC()

　- もし… IF文

　- GOTO @ラベル名

　- SAVE 110 で保存

その他のクリスマス・ソング 

【注意】他の曲を消さないとメモリオーバーで入りません 

♪きよしこの夜 
<G.A8GE2. G.A8GE2R <D2D>B2. <C2C>G2R 

A2A<C.>B8A G.A8GE2R A2A<C.>B8A G.A8GE2R 

<D2DF.D8>B<C2.E2R C>GEG.F8DC2.C2R 

♪赤鼻のトナカイ 
T180<GA8.G16E<C>AG2R G8.A16G8.A16G<C>B2.R 

F8.G16FDBAG2R G8.A16G8.A16GAE2.R 

G8.A16GE<C>AG2R G8.A16G8.A16G<C>B2.R 

FG8.F16DBAG2R G8.A16G8.A16G<DC2.R 

◆数字キーをシンプルにあつかおう 

―――――――――――――――――――― 
1 'JukeBox 2 

10 A="T60 <BGAD2.R DABG2.R" 

20 B=“T180<G.G8A2G2<C2>B2.R 

G.G8A2G2<D2C2.>RG.G8<G2E2C. 

C8>B2A2.R <F.F8E2C2D2C2.R" 

30C=“T200<D#8C#8>F#8R8<F#F# 

D#8C#8>F#8R8<F#F#D#8C#8>F#8 

R8<F#>D#<F#>C#<FFD#8C#8>C#8 

R8<FFD#8C#8>C#8R8<FFD#8C#8 

>C#<F>D#<F>F#<F#F#" 

40　D=“T180<EEE2EEE2EGC.D8E2 

R2FFF.F8FEE.E8EDDCD2GR" 

50 E=“<G<CC8D8C8>B8AAA<DD8 

E8D8C8>BGG<EE8F8E8D8C>AG8 

G8A<D>B<C" 

60 F=“<<C>B8.A16G.F8EDC.G8 

A.A8B.B8<C.” 

100 Z=ASC("0"):CLS 

110 ?"Ǯ1 ﾁｬｲﾑ” 

120 ?"Ǯ2 ﾊｯﾋﾟｰﾊﾞｰｽﾃﾞｰ" 

130 ?"Ǯ3 ﾈｺ ﾌﾝｼﾞｬｯﾀ....” 

140 ?"Ǯ4 ｼﾞﾝｸﾞﾙ ﾍﾞﾙ" 

150 ?"Ǯ5 ﾒﾘｸﾘ ｵﾒﾃﾞﾄ" 

160 ?"Ǯ6 ﾓﾛﾋﾞﾄ ｺｿﾞﾘﾃ"  

200 @SONG: K=INKEY(): K=K-Z 

210 IF K=1 PLAY A 

220 IF K=2 PLAY B 

230 IF K=3 PLAY C 

240 IF K=4 PLAY D 

250 IF K=5 PLAY E 

260 IF K=6 PLAY F 

300 WAIT 5: GOTO @SONG 

――――――――――――――――――――

曲データ

Zに"０"の⽂字コードを代⼊

⽂字コードを１〜６の数値に変換



 

  /  13 40

　 
　- ASC("0") = 48

　

　- 0（数） と "0"（数字） はちがう

　

　- K=K-Z （上書き代入）

　

　- SAVE 110 で保存



 

  /  14 40

３．電子ピアノ 

 ◆１２３でドレミ
――――――――――――――――――――
1 'Piano 0 

110 @LOOP 

120 A=INKEY() 

240 IF A=ASC("1") PLAY "C" 

250 IF A=ASC("2") PLAY "D" 

260 IF A=ASC("3") PLAY "E" 

350 GOTO @LOOP 

――――――――――――――――――――

押されたキーを読み取り、Aに代⼊

ドの⾳を鳴らす
レの⾳を鳴らす

ミの⾳を鳴らす

@LOOPへ⾏く

　- NEW してプログラム入力

　

　- ASC("数字") が面倒…

　

　- SAVE 111 で保存

 ◆数字キーをシンプルにあつかう

――――――――――――――――――――
1 'Piano 1 

100 Z=ASC("0") 

110 @LOOP 

120 A=INKEY() 

150 B=A-Z 

240 IF B=1 PLAY "C" 

250 IF B=2 PLAY "D" 

260 IF B=3 PLAY "E" 

270 IF B=4 PLAY "F" 

280 IF B=5 PLAY "G" 

290 IF B=6 PLAY "A" 

300 IF B=7 PLAY "B" 

310 IF B=8 PLAY "<C" 

350 GOTO @LOOP

――――――――――――――――――――

Zに"０"の⽂字コードを代⼊

⽂字コードを１〜６の数値に変換、Bに代⼊

ファの⾳を鳴らす
ソの⾳を鳴らす
ラの⾳を鳴らす
シの⾳を鳴らす
⾼いドの⾳を鳴らす

　- B=A-Z （別の変数に代入）

　

　- SAVE 111 で保存

◆けんばんを見ながら♪

――――――――――――――――――――
1 'Piano 2 

10 CLS: VIDEO 2 

20 ?"I F F I F F F I F F I" 

30 ?"I F F I F F F I F F I" 

40 ?"I I I I I I I I I I I" 

50 ?"I1I2I3I4I5I6I7I8I9I0I" 

60 ?"QHMHMHMHMHMHMHMHMHMHR" 

70 ?" ﾄﾞﾚ ﾐ ﾌｧｿ ﾗ ｼ ﾄﾞﾚ ﾐ" 

100 Z=ASC("0") 

110 @LOOP 

120 A=INKEY() 

150 B=A-Z 

240 IF B=1 PLAY "<C" 

250 IF B=2 PLAY "<D" 

260 IF B=3 PLAY "<E" 

270 IF B=4 PLAY "<F" 

280 IF B=5 PLAY "<G" 

290 IF B=6 PLAY "<A" 

300 IF B=7 PLAY "<B" 

310 IF B=8 PLAY "<<C" 

320 IF B=9 PLAY "<<D" 

330 IF B=0 PLAY "<<E" 

350 GOTO @LOOP

――――――――――――――――――――

画⾯の⾊を⽩⿊反転

記号を使ってけんばんを描く

⾼いレの⾳を鳴らす
⾼いミの⾳を鳴らす

オクターブ上げ

　- 絵文字でけんばんを描こう

　 

　- VIDEO 2 で白黒反転 （VIDEO 1 で元通り）

　 

　- オクターブを上げよう

　 

　- 鳴る音をふやそう

　 

　- SAVE 111 で保存

◆けんばんに♪を表示させよう

――――――――――――――――――――
1 'Piano 3 

10 CLS: VIDEO 2 

20 ?"I F F I F F F I F F I" 

30 ?"I F F I F F F I F F I" 

40 ?"I I I I I I I I I I I" 

50 ?"I1I2I3I4I5I6I7I8I9I0I" 

60 ?"QHMHMHMHMHMHMHMHMHMHR" 

70 ?" ﾄﾞﾚ ﾐ ﾌｧｿ ﾗ ｼ ﾄﾞﾚ ﾐ" 

100 A=0: X=1: Y=2: Z=ASC("0") 

110 @LOOP: B=A: C=0 

120 @KEY: A=INKEY() 

130 IF A=0 C=C+1: IF C<72 GOTO @KEY 

140 IF A<>B LC X,Y: ?" " 

150 B=A-Z: IF B<0 OR B>9 GOTO @LOOP 

240 IF B=1 PLAY "<C" 

250 IF B=2 PLAY "<D" 

260 IF B=3 PLAY "<E" 

270 IF B=4 PLAY "<F" 

280 IF B=5 PLAY "<G" 

290 IF B=6 PLAY "<A" 

300 IF B=7 PLAY "<B" 

310 IF B=8 PLAY "<<C" 

320 IF B=9 PLAY "<<D" 

330 IF B=0 PLAY "<<E": B=10 

340 X=B*2-1: LC X,Y: ?"e" 

350 GOTO @LOOP

――――――――――――――――――――

座標⽤変数X,Yを定義

前回押したキーと、今回を⽐較

↑数字キーからの⼊⼒のみを許す

座標X,Yに♪を表⽰
０は１０として座標を計算する

　 - 座標(X,Y)指定命令 LOCATE / LC

　

　- キー入力なければ INKEY() は 0 を返す

　 

　- A<>B … AがBと異なるか （A!=B とも書ける）

　 

　- B<0 OR B>9 … Bが0より小さいか、または、9より大きいか

　 

　- X=B*2-1 （B=1,2,3… だったら X=1,3,5…）

　 

　- SAVE 111 で保存



 

  /  22 40

４．リプレイ（順番当てゲーム）

 ◆サイコロを振りたまえ… 乱数 RND()

 ◆102個の変数… 配列 [0], [1], [2], …

 ◆サブルーチン… GOSUB と RETURN文

 ◆くり返し… FOR ～ NEXT文
――――――――――――――――――――
1 'Replay 0 

10 CLS: VIDEO 2 

20 ?"I F F I F F F I F F I" 

30 ?"I F F I F F F I F F I" 

40 ?"I I I I I I I I I I I" 

50 ?"I1I2I3I4I5I6I7I8I9I0I" 

60 ?"QHMHMHMHMHMHMHMHMHMHR" 

70 ?" ﾄﾞﾚ ﾐ ﾌｧｿ ﾗ ｼ ﾄﾞﾚ ﾐ" 

100 S=0: Y=2 

110 @QUIZ: WAIT 120 

120 [S]=RND(8)+1 

130 FOR C=0 TO S: B=[C]: GOSUB @SUB: NEXT 

210 S=S+1: GOTO @QUIZ 

230 @SUB 

240 IF B=1 PLAY "<C" 

250 IF B=2 PLAY "<D" 

260 IF B=3 PLAY "<E" 

270 IF B=4 PLAY "<F" 

280 IF B=5 PLAY "<G" 

290 IF B=6 PLAY "<A" 

300 IF B=7 PLAY "<B" 

310 IF B=8 PLAY "<<C" 

340 X=B*2-1: LC X,Y: ?"e" 

350 WAIT 30: LC X,Y: ?" " 

360 RETURN

――――――――――――――――――――

S：問題の⻑さ Y：縦の位置 

配列のS番⽬にランダムな数を追加

↑Cが０からSに代⼊されてる数になるまで、NEXTまでの間の処理をくり返す
Cの値は１ずつふえる

問題の⻑さを１問ふやす

0.5秒後に♪マークを消す
GOSUBの呼び出し元に戻る

消す行　次の通り入力 

140 

150 

320 

330

　- ? RND(6) で変なサイコロ…

　

　- [0]、[1]、…[101] も変数として使える！

　

　- LOAD 111 してプログラム入力

　

　- どんどん覚えて… [S]=RND(8)+1

　

　- 1ふやす S=S+1

　

　- FOR ～ NEXT の間をぐるぐる… 

　

　- 必ず帰ってきてね… GOSUB 行番号/@ラベル名 と 

　　 RETURN（または GSB と RTN）

　

　- 音が無限にふえる…？

　

　- SAVE 112 で保存

 ◆答えをチェックしよう 
―――――――――――――――――――― 

1 'Replay 1 

10 CLS: VIDEO 2 

20 ?"I F F I F F F I F F I" 

30 ?"I F F I F F F I F F I" 

40 ?"I I I I I I I I I I I" 

50 ?"I1I2I3I4I5I6I7I8I9I0I" 

60 ?"QHMHMHMHMHMHMHMHMHMHR" 

70 ?" ﾄﾞﾚ ﾐ ﾌｧｿ ﾗ ｼ ﾄﾞﾚ ﾐ" 

100 S=0: Y=2: Z=ASC("0") 

110 @QUIZ: WAIT 120 

120 [S]=RND(8)+1: CLK 

130 FOR C=0 TO S: B=[C]: GOSUB @SUB: NEXT 

140 FOR C=0 TO S 

150 @KEY 

160  A=INKEY(): IF A=0 GOTO @KEY 

170  B=A-Z 

180  GOSUB @SUB 

190  IF B<>[C] END 

200 NEXT 

210 S=S+1: GOTO @QUIZ 

230 @SUB 

240 IF B=1 PLAY "<C" 

250 IF B=2 PLAY "<D" 

260 IF B=3 PLAY "<E" 

270 IF B=4 PLAY "<F" 

280 IF B=5 PLAY "<G" 

290 IF B=6 PLAY "<A" 

300 IF B=7 PLAY "<B" 

310 IF B=8 PLAY "<<C" 

340 X=B*2-1: LC X,Y: ?"e" 

350 WAIT 30: LC X,Y: ?" " 

360 RETURN

――――――――――――――――――――

Zに"0"の⽂字コードを代⼊

キーバッファクリア

Cが０からSに代⼊されてる
数になるまで、この間の処理
をくり返す 

Cの値は１ずつふえる

　- キーバッファとクリア命令 CLK

　

　- 間違えたら終わりだ！ IF B<>[C] END

　

　- デタラメなキーを押したら…変だぞ？

　

　- SAVE 112 で保存

◆もし、デタラメなキーを押したら…許す！
――――――――――――――――――――
1 'Replay 2 

10 CLS: VIDEO 2 

20 ?"I F F I F F F I F F I" 

30 ?"I F F I F F F I F F I" 

40 ?"I I I I I I I I I I I" 

50 ?"I1I2I3I4I5I6I7I8I9I0I" 

60 ?"QHMHMHMHMHMHMHMHMHMHR" 

70 ?" ﾄﾞﾚ ﾐ ﾌｧｿ ﾗ ｼ ﾄﾞﾚ ﾐ" 

100 S=0: Y=2: Z=ASC("0") 

110 @QUIZ: WAIT 120 

120 [S]=RND(8)+1: CLK 

130 FOR C=0 TO S: B=[C]: GOSUB @SUB: NEXT 

140 FOR C=0 TO S 

150 @KEY 

160  A=INKEY(): IF A=0 GOTO @KEY 

170  B=A-Z: IF B<1 OR B>8 GOTO @KEY 

180  GOSUB @SUB 

190  IF B<>[C] END 

200 NEXT 

210 S=S+1: GOTO @QUIZ 

230 @SUB 

240 IF B=1 PLAY "<C" 

250 IF B=2 PLAY "<D" 

260 IF B=3 PLAY "<E" 

270 IF B=4 PLAY "<F" 

280 IF B=5 PLAY "<G" 

290 IF B=6 PLAY "<A" 

300 IF B=7 PLAY "<B" 

310 IF B=8 PLAY "<<C" 

340 X=B*2-1: LC X,Y: ?"e" 

350 WAIT 30: LC X,Y: ?" " 

360 RETURN

――――――――――――――――――――

↑数字の１〜８以外のキーは無視する

　- IF B<1 OR B>8 … もし、Bが1より小さいか、 

　　または、8より大きければ…

　

　- なんかさびしい…

　

　- SAVE 112 で保存

◆ゲームっぽくしよう
――――――――――――――――――――
1 'Replay 3 

10 CLS: VIDEO 2 

20 ?"I F F I F F F I F F I" 

30 ?"I F F I F F F I F F I" 

40 ?"I I I I I I I I I I I" 

50 ?"I1I2I3I4I5I6I7I8I9I0I" 

60 ?"QHMHMHMHMHMHMHMHMHMHR" 

70 ?" ﾄﾞﾚ ﾐ ﾌｧｿ ﾗ ｼ ﾄﾞﾚ ﾐ" 

100 S=0: Y=2: Z=ASC("0") 

110 @QUIZ: WAIT 120 

120 [S]=RND(8)+1: CLK 

130 FOR C=0 TO S: B=[C]: GOSUB @SUB: NEXT 

140 FOR C=0 TO S 

150 @KEY 

160  A=INKEY(): IF A=0 GOTO @KEY 

170  B=A-Z: IF B<1 OR B>8 GOTO @KEY 

180  GOSUB @SUB 

190  IF B<>[C] PLAY ">B": LC 0,7: ?"SCORE=";S: END 

200 NEXT 

210 PLAY "<<E8C": S=S+1: GOTO @QUIZ 

230 @SUB 

240 IF B=1 PLAY "<C" 

250 IF B=2 PLAY "<D" 

260 IF B=3 PLAY "<E" 

270 IF B=4 PLAY "<F" 

280 IF B=5 PLAY "<G" 

290 IF B=6 PLAY "<A" 

300 IF B=7 PLAY "<B" 

310 IF B=8 PLAY "<<C" 

340 X=B*2-1: LC X,Y: ?"e" 

350 WAIT 30: LC X,Y: ?" " 

360 RETURN

――――――――――――――――――――

↑ブー             座標(0,7)にスコア表⽰            S:スコア変数

↑ピンポン



 

  /  29 40

　 

　- 効果音で演出しよう

　 

　- ゲームオーバーでスコア表示

　 

　- SAVE 112 で保存

５．ピアノの達人（キー押しゲーム）

 ◆？で自動スクロール
――――――――――――――――――――
1 'Tatsujin 0 

10 CLS: VIDEO 2 

20 ?"I F F I F F F I F F I" 

30 ?"I F F I F F F I F F I" 

40 ?"I I I I I I I I I I I" 

50 ?"I1I2I3I4I5I6I7I8I9I0I" 

60 ?"QHMHMHMHMHMHMHMHMHMHR" 

70 ?" ﾄﾞﾚ ﾐ ﾌｧｿ ﾗ ｼ ﾄﾞﾚ ﾐ" 

120 @LOOP 

150 B=RND(8)+1: X=B*2-1: LC X,23: ?B; 

160 WAIT 30 

180 GOSUB @SUB 

220 GOTO @LOOP 

230 @SUB 

240 IF B=1 PLAY "<C" 

250 IF B=2 PLAY "<D" 

260 IF B=3 PLAY "<E" 

270 IF B=4 PLAY "<F" 

280 IF B=5 PLAY "<G" 

290 IF B=6 PLAY "<A" 

300 IF B=7 PLAY "<B" 

310 IF B=8 PLAY "<<C" 

340 LC X,23: ?"e" 

350 WAIT 30 

360 RETURN

――――――――――――――――――――

↑Bにランダムな数を代⼊、 対応する座標に数字を表⽰

数字を♪マークに変換

消す行　次の通り入力 

100 

110 

130 

140 

170 

190 

200 

210

　- LOAD 112 してプログラム入力

　　

　- 最下行(Y=23)で PRINT / ? すると…スクロール？

　 
　- SAVE 113 で保存

◆答えをチェックしよう
――――――――――――――――――――
1 'Tatsujin 1 

10 CLS: VIDEO 2 

20 ?"I F F I F F F I F F I" 

30 ?"I F F I F F F I F F I" 

40 ?"I I I I I I I I I I I" 

50 ?"I1I2I3I4I5I6I7I8I9I0I" 

60 ?"QHMHMHMHMHMHMHMHMHMHR" 

70 ?" ﾄﾞﾚ ﾐ ﾌｧｿ ﾗ ｼ ﾄﾞﾚ ﾐ" 

100 S=0: Y=0: Z=ASC("0") 

120 @LOOP 

140 LC 19,Y: ?"SCORE=";S 

150 B=RND(8)+1: X=B*2-1: LC X,23: ?B; 

160 FOR C=1 TO 20: WAIT 1 

170  A=INKEY(): B=A-Z: X=B*2-1 

180  IF A>0 AND SCR(X,Y)=A GOSUB @SUB 

190 NEXT 

210 LC 0,23: ? 

220 GOTO @LOOP 

230 @SUB 

240 IF B=1 PLAY "<C" 

250 IF B=2 PLAY "<D" 

260 IF B=3 PLAY "<E" 

270 IF B=4 PLAY "<F" 

280 IF B=5 PLAY "<G" 

290 IF B=6 PLAY "<A" 

300 IF B=7 PLAY "<B" 

310 IF B=8 PLAY "<<C" 

340 LC X,Y: ?"e" 

350 S=S+1 

360 RETURN

――――――――――――――――――――

S:スコア   Y:縦位置   Z:"０"の⽂字コード

座標(19,Y)にスコア表⽰

スクロール

キー⼊⼒待ちと 
正解判定

正解したら数字を♪に変更
スコアを１ポイントふやす

　- IF A>0 AND SCR(X,Y)=A … もし、 

　　Aが0より大きくて、かつ、(X,Y)の文字コードがAならば…

　 

　- LC 0,23: ? でスクロール

　 

　- スコアを表示しよう

　 

　- むずい！ クソゲー！ いつ終わるの？

　 

　- SAVE 113 で保存

◆タイマー制御
――――――――――――――――――――
1 'Tatsujin 2 

10 CLS: VIDEO 2 

20 ?"I F F I F F F I F F I" 

30 ?"I F F I F F F I F F I" 

40 ?"I I I I I I I I I I I" 

50 ?"I1I2I3I4I5I6I7I8I9I0I" 

60 ?"QHMHMHMHMHMHMHMHMHMHR" 

70 ?" ﾄﾞﾚ ﾐ ﾌｧｿ ﾗ ｼ ﾄﾞﾚ ﾐ" 

100 S=0: Y=0: Z=ASC("0") 

110 CLT 

120 @LOOP: Q=TICK() 

130 LC 28,0: ?Q 

140 LC 19,Y: ?"SCORE=";S 

150 B=RND(24)+1: X=B*2-1: LC X,23: IF B<9 ?B; 

160 FOR C=1 TO 20: WAIT 1 

170  A=INKEY(): B=A-Z: X=B*2-1 

180  IF A>0 AND SCR(X,Y)=A GOSUB @SUB 

190 NEXT 

200 IF Q>=3600 LC 0,22: END 

210 LC 0,23: ? 

220 GOTO @LOOP 

230 @SUB 

240 IF B=1 PLAY "<C" 

250 IF B=2 PLAY "<D" 

260 IF B=3 PLAY "<E" 

270 IF B=4 PLAY "<F" 

280 IF B=5 PLAY "<G" 

290 IF B=6 PLAY "<A" 

300 IF B=7 PLAY "<B" 

310 IF B=8 PLAY "<<C" 

340 LC X,Y: ?"e" 

350 S=S+1 

360 RETURN

――――――――――――――――――――

タイマーリセット

座標(28,0) にタイマー時間表⽰

↑ 数字1〜８の出現確率を1/3に下げる  ↑

６０秒でタイムアップ

　-タイマー命令 CLT でリセット、TICK() 

　　で時間(1秒で60)を返す

　- 1分で終わりだ！ IF Q>=3600 … END

　- 出現確率を下げる… RND(24)

　- SAVE 113 で保存

◆カウントダウンタイマー表示

 ◆部分スクロール（中級）

――――――――――――――――――――

1 'Tatsujin 3 

10 CLS: VIDEO 2 

20 ?"I F F I F F F I F F I" 

30 ?"I F F I F F F I F F I" 

40 ?"I I I I I I I I I I I" 

50 ?"I1I2I3I4I5I6I7I8I9I0I" 

60 ?"QHMHMHMHMHMHMHMHMHMHR" 

70 ?" ﾄﾞﾚ ﾐ ﾌｧｿ ﾗ ｼ ﾄﾞﾚ ﾐ" 

80 ?" 3 3 3 3 3 3 3 3" 

100 S=0: Y=7: Z=ASC("0") 

110 D=#900+Y*32: CLT 

120 @LOOP: Q=TICK()/60 

130 LC 30,0: ?DEC$(60-Q,2) 

140 LC 21,Y: ?"SCORE=";S 

150 B=RND(24)+1: X=B*2-1: LC X,22: IF B<9 ?B; 

160 FOR C=1 TO 20: WAIT 1 

170  A=INKEY(): B=A-Z: X=B*2-1 

180  IF A>0 AND SCR(X,Y)=A GOSUB @SUB 

190 NEXT 

200 IF Q>=60 LC 0,22: END 

210 COPY D,D+32,(23-Y)*32 

220 GOTO @LOOP 

230 @SUB 

240 IF B=1 PLAY "<C" 

250 IF B=2 PLAY "<D" 

260 IF B=3 PLAY "<E" 

270 IF B=4 PLAY "<F" 

280 IF B=5 PLAY "<G" 

290 IF B=6 PLAY "<A" 

300 IF B=7 PLAY "<B" 

310 IF B=8 PLAY "<<C" 

340 LC X,Y: ?"e" 

350 S=S+1 

360 RETURN

――――――――――――――――――――

Dにスクロール開始アドレスを代⼊
１秒単位に変換 
秒数２けた表⽰ 

部分スクロール処理 

　- あと何秒　 ? 60 - TICK()/60

　- DEC$(数,2) で２けた右そろえ表示

　- (ビデオ)メモリコピー命令 COPY で部分スクロール

　- SAVE 113 で保存

６．アプリ・セレクター
 ◆ロード後実行命令 LRUN
――――――――――――――――――――
1 'Selector & JukeBox 

10 A="T60 <BGAD2.R DABG2.R" 

20 B="T180<G.G8A2G2<C2>B2.R G.G8A2G2<D2C2.>R  

 G.G8<G2E2C.C8>B2A2.R <F.F8E2C2D2C2.R" 

30 C="T200<D#8C#8>F#8R8<F#F# D#8C#8>F#8R8<F#F# 

 D#8C#8>F#8R8<F#>D#<F#>C#<FF D#8C#8>C#8R8<FF  

 D#8C#8>C#8R8<FF D#8C#8>C#<F>D#<F>F#<F#F#" 

40 D="T180<EEE2EEE2 EGC.D8E2R2 FFF.F8FEE.E8 EDDCD2GR" 

50 E="<G<CC8D8C8>B8AAA<DD8E8D8C8>BGG<EE8F8E8D8C>AG8G8A<D>B<C" 

60 F="<<C>B8.A16G.F8EDC.G8 A.A8B.B8<C.R8" 

100 Z=ASC("0"): GOSUB @SELECT: CLS 
110 ?"e1 ﾁｬｲﾑ" 

120 ?"e2 ﾊｯﾋﾟｰﾊﾞｰｽﾃﾞｰ" 

130 ?"e3 ﾈｺ ﾌﾝｼﾞｬｯﾀ" 

140 ?"e4 ｼﾞﾝｸﾞﾙ ﾍﾞﾙ" 

150 ?"e5 ﾒﾘｸﾘ ｵﾒﾃﾞﾄ" 

160 ?"e6 ﾓﾛﾋﾞﾄ ｺｿﾞﾘﾃ" 

200 @SONG: K=INKEY(): K=K-Z 

210 IF K=1 PLAY A 

220 IF K=2 PLAY B 

230 IF K=3 PLAY C 

240 IF K=4 PLAY D 

250 IF K=5 PLAY E 

260 IF K=6 PLAY F 

300 WAIT 5: GOTO @SONG 

400 @SELECT: CLS 

410 ?"Select App:" 

420 ?"1 ｼﾞｭｰｸ･ﾎﾞｯｸｽ" 

430 ?"2 ﾋﾟｱﾉ" 

440 ?"3 ﾋﾟｱﾉ ﾘﾌﾟﾚｲ" 

450 ?"4 ﾋﾟｱﾉ ﾉ ﾀﾂｼﾞﾝ" 

460 @APP: K=INKEY(): K=K-Z 

470 IF K=1 RETURN 

480 IF K=2 LRUN 111 

490 IF K=3 LRUN 112 

500 IF K=4 LRUN 113 

510 WAIT 5: GOTO @APP

――――――――――――――――――――

EEPROM 111番のプログラムを読み込み実⾏ 
EEPROM 112番のプログラムを読み込み実⾏ 
EEPROM 113番のプログラムを読み込み実⾏ 

　- LOAD 110 してプログラム入力

　- GOSUB / GSB でアプリ選択メニューを作ろう

　- RETURN / RTN でジューク・ボックスにもどる

　- LRUN 111～113 で他アプリをロード＆実行

　- SAVE 110 で保存

７．おわりに

 ◆次回開催予定

　- 2019/3/21(木・祝) … 場所・内容は「ひ・み・つ…」

ありがとうございました．またね！


